
EC362 Lab #1 

SPIM Starter Guide 

SPIM (MIPS spelled backwards) 

SPIM is a simulator that provides a way of writing assembly-level programs that can be 
translated to MIPS instructions that can then be executed. By using SPIM we can study at a 
fundamental level the use of registers, memory, and system I/O in a modern architecture. If you 
want to understand how a computer really works, a good place to start is by observing how its 
components are used to store and execute a program. SPIM can help us visualize the workings 
of a processor like the MIPS, as well as its need for memory and primitive I/O. 

SPIM executes as a Windows program and includes windows that display 

• Messages– notes from SPIM that usually are concerned with success/failure of our use 
of SPIM. 

• Registers– the contents of the 32 registers that are in a MIPS processor. We will ignore 
the additional 32 floating point registers. 

• Data– memory used to store data on which our program is working. 
• Text– memory used to store the MIPS program that is executing. Note that we will write 

SPIM programs, but they will be translated to their corresponding MIPS code. 
• Console– I/O for our program. 

We will write our programs using an editor like wordpad, saving them in files with a .asm suffix. 
The SPIM interface will then let us load the file we have created, although it will translate the 
program to MIPS code before the program is actually loaded into the text segment of memory. 
Once the program is loaded, we will be able to run the program and observe the changes in the 
registers and data segment in memory. For instance, we will be able to start the program and 
execute one instruction at a time, a particularly useful way of understanding how the MIPS 
machine works. 

The MIPS processor contains 32 general-purpose registers that are used during execution of a 
program. Here are the more common ones (unless otherwise specified, all registers must be 
preserved across function calls): 

• $zero (register 0) always contains a 0 
• $v0..$v1 (registers 2-3) used in making system calls (e.g., I/O) and in returning values 

from functions (need not be preserved across function calls) 
• $a0..$a4 (registers 4-7) used in passing arguments (parameters) to functions 
• $t0..$t7 (registers 8-15) temporary registers that are not preserved across function calls 
• $s0..$s7 (registers 16-23) temporary registers that are preserved across function calls 
• $t8..$t9 (registers 24-25) temporary registers (not preserved across function calls) 



• $gp (register 28) stores the global pointer for accessing static data 
• $sp (register 29) stores the stack pointer 
• $fp (register 30) stores the frame pointer for accessing activation records 
• $ra (register 31) stores the return address for function calls 

In the Register window, there are several other registers to which we will need to pay 
attention. The Program Counter (PC) contains the address of the next instruction to be 
executed as the program runs. The Status register contains a set of values that can be tested to 
determine the status of execution. 

MIPS has a well-designed instruction set. However, the instructions are very primitive. Here are 
just a few of the common instructions we will use.  

add Dest, Src1, Src2 
addi Dest, Src, Imm 
Puts the sum of integers from register Src1 and Src2 or Imm into register Dest. 

sub Dest, Src1, Src2 
Puts the difference of integers from register Src1 and Src2 into register Dest. 

beq Src1, Src2, label 
Branch to label if register Src1 equals Src2. 

bne Src1, Src2, label 
Branch to label if register Src1 does not equal Src2. 

j label 
Always (unconditionally) jump to address specified by label. 

lw Dest, address 
Load the word (32-bits) at address into register Dest. 

sw Src, address 
Store the word from register Src at address. 

SPIM also has a set of system calls that can be used in it. Those calls are typically used for input 
and output, or to stop execution of the program. More on these later. 

Let's try to put this in perspective by examining a SPIM program to compute the sum of the 
integers from 1 to n, the answer to which is n*(n+1) / 2. 

  



# MIPS example for computing 1 + 2 + … + n 
 
 .data 
prompt: .asciiz "Enter n: " 
endl: .asciiz "\n" 
result: .asciiz "1 + 2 +    + " 
equals: .asciiz " = " 
  
 .text 
main: addi $v0, $zero, 4  # print string syscall number 
 la $a0, prompt  # address of prompt 
 syscall    # print prompt 
 
 addi $v0, $zero, 5  # read integer syscall number 
 syscall    # read it (result is in $v0) 
 
 addi $t1, $v0, 0  # save n for later 
 addi $s0, $v0, 0   # initialize sum = n 
  
 addi $t0, $zero, 1  # set count = 1 
loop: beq $t0, $t1, done  # count = =  n? 
 add $s0, $s0, $t0  # sum = sum + count 
 addi $t0, $t0, 1  # count++ 
 j loop   # repeat 
 
done: addi $v0, $zero, 4  # print string syscall 
 la $a0, result   # prepare to print header of result 
 syscall    # print it 
 
 addi $v0, $zero, 1  # print integer syscall 
 addi $a0, $t1, 0   # copy n into a0 
 syscall    # print n 
 
 addi $v0, $zero, 4  # print string syscall 
 la $a0, equals   # prepare to print equals 
 syscall    # print it 
 
 addi $v0, $zero, 1  # print integer syscall 
 addi $a0, $s0, 0   # copy sum into a0 
 syscall    # print it 
 
 addi $v0, $zero, 4  # print string syscall 
 la $a0, endl   # prepare to print endl 
 syscall    # print it 



  
 addi $v0, $zero, 10  # exit syscall number 
 syscall    # exit 
 

All text that follows a # is a comment – a statement a programmer makes to help a reader 
understand a portion of the program. 

The first section of our programs will normally be the data segment, containing the static data 
that we wish to use in the program. In this program, prompt is an ASCII Z-string (a string 
terminated by \0). So are endl and result and equals. If we had wanted to store an integer 
(word) such as 5 in a memory position labeled by n, you'd just say "n: .word 5". 

The next section of our programs will normally be the text segment, containing the instructions 
of our program. 

Within the text segment there are groups of instructions that we will use often. For instance, to 
print the string "Enter n: "stored at the address "prompt", the instructions 

 addi $v0, $zero, 4  # print string syscall number 
 la $a0, prompt  # address of prompt 
 syscall    # print prompt 
 
are used. When we load 4 into the $v0 register, we are indicating that we want to print a string 
on the Console. Load 1 if we want to print an integer, 5 if we want to read an integer, and 8 if 
we want to read a string. The la instruction specifies the address of the information to be 
printed. Finally, the SPIM instruction syscall causes the system call to occur and the string will 
be printed on the console. 

Please note that an explicit action should be taken to terminate execution of the program. 
When we load 10 into the v0 register, then make the system call, that is interpreted as halt by 
SPIM. 

In MIPS (and therefore SPIM), memory is byte addressable. That means that each byte of 
memory can be referenced by a MIPS instruction. That does NOT mean that all values that we 
want to use must be stored in a byte. For instance, common representations for integers and 
floating point values use 32 bits (4 bytes or 1 word). Each character is stored in a byte. Keep this 
in mind as you observe the data segment window. 

Each instruction is also contained in a word. When you open the text segment window, you will 
see four sets of information. On the right you will identify instructions from the SPIM program 
you are running. The numbers correspond to the line numbers of the SPIM program. The next 
set of information to the left is the corresponding MIPS instructions. Note that one SPIM 
instruction is not always assembled as one MIPS instruction. A single MIPS instruction can 



become 1, 2, or 3 MIPS instructions, just like pseudo-instructions. The next column is the actual 
MIPS instructions. This is important for you to recognize! The last column (the leftmost column) 
is the address of each instruction. Note that the addresses increase by 4 since each instruction 
occupies 4 bytes. 

Lab part 1 – use the MIPS code above to get acquainted with SPIM.  Load the MIPS code for 
sum, run it, step thru it line by line, observe the register contents as they change, etc.  Rerun it 
again.  Get familiar with SPIM!! 

Lab part 2 – the following MIPS program tries to copy words from memory at the address in register 
$a0 to memory at the address in register $a1, keeping count of the number of words copied in register 
$v0. The program stops copying when it finds a word equal to 0, and this value should not be included in 
the count. The contents of registers $v1, $a0, and $a1 are not (and do not have to be) preserved.  

 loop: lw $v1, 0($a0)  
 addi $v0, $v0, 1  
 sw $v1, 0($a1)  
 addi $a0, $a0, 1  
 addi $a1, $a1, 1  
 bne $v1, $zero, loop  
 
There are multiple bugs in this MIPS program; use spim to identify the errors, and then fix them by 
writing a bug-free version.  Be sure to test your working version using spim! 

